

Formalization of Gyrovector Spaces as Models of Hyperbolic Geometry and Special Relativity

Jelena Marković^{1,2*} and Filip Marić^{1,3*}

^{1*} Computer Science Department, Faculty of Mathematics, Studentski Trg 16, Belgrade, 11000, Serbia, Serbia.

²ORCID: 0009-0007-4797-2666.

³ORCID: 0000-0001-7219-6960.

*Corresponding author(s). E-mail(s): jelena.markovic@matf.bg.ac.rs; filip.maric@matf.bg.ac.rs;

Abstract

In this paper, we present an Isabelle/HOL formalization of non-commutative and non-associative algebraic structures known as *gyrogroups* and *gyrovector spaces*. These concepts were introduced by Abraham A. Ungar and have deep connections to hyperbolic geometry and special relativity. Gyrovector spaces can be used to define models of hyperbolic geometry. Unlike other models, gyrovector spaces offer the advantage that all definitions exhibit remarkable syntactical similarities to standard Euclidean and Cartesian geometry (e.g., points on the line between \mathbf{a} and \mathbf{b} satisfy the parametric equation $\mathbf{a} \oplus \mathbf{t} \otimes (\ominus \mathbf{a} \oplus \mathbf{b})$, for $\mathbf{t} \in \mathbb{R}$, while the hyperbolic Pythagorean theorem is expressed as $\mathbf{a}^2 \oplus \mathbf{b}^2 = \mathbf{c}^2$, where \otimes , \oplus , and \ominus represent gyro operations).

We begin by formally defining gyrogroups and gyrovector spaces and proving their numerous properties. Next, we formalize Möbius and Einstein models of these abstract structures, and then demonstrate that these are equivalent to the Poincaré and Klein-Beltrami models, satisfying Tarski's geometry axioms for hyperbolic geometry.

Keywords: gyrogroup, gyrovector space, Poincaré Disc, Möbius gyrovector space, hyperbolic geometry, theory of relativity

1 Introduction

Non-Euclidean geometries, including hyperbolic geometry, have been studied since the 19th century, and their properties are now well understood. Applications of hyperbolic geometry in physics have also been widely explored, with findings indicating that hyperbolic geometry provides the mathematical foundation for the theory of special relativity, just as Euclidean geometry underpins classical Newtonian and Galilean mechanics.

Analytic definitions of hyperbolic geometry are typically framed in the extended complex plane, heavily relying on linear algebra techniques (such as Hermitian matrices) and hyperbolic trigonometry [1]. Unlike the Cartesian approach used in Euclidean geometry, hyperbolic geometry does not incorporate vectors in the same manner. In Euclidean geometry, vectors form a natural framework: vector addition is associative and commutative, forming an Abelian group, and vectors can be scaled by real numbers, creating a vector space. The dot (inner) product and vector norm are easily defined, establishing the Euclidean metric, $d(A, B) = |\overrightarrow{AB}|$. However, these concepts are not as straightforward to define in the hyperbolic plane, leading to analytic definitions of hyperbolic geometry that do not employ vector spaces. This results in an inherent asymmetry between the definitions of Euclidean and hyperbolic geometry.

In 1988, Ungar [2] proposed an alternative and developed algebraic structures known as *gyrogroups* and *gyrovector spaces*, enabling the definition and use of "vectors" for the formalization of hyperbolic geometry. Although operations with these "vectors" do not form traditional vector spaces-vector addition, for instance, is neither commutative nor associative-the elements, called gyrovectors, along with their associated operations, satisfy enough algebraic properties to allow a formalism of hyperbolic geometry that is strikingly similar in syntax to the classical treatment of Euclidean geometry.

Interestingly, these structures were first discovered through an examination of Einstein's velocity addition laws, which were then unified with classical Galilean velocity addition. This unification revealed a deep connection between the special theory of relativity and hyperbolic geometry. The experimentally discovered Thomas precession (also known as Thomas gyration) was identified as a "missing link" between Einstein's velocity addition formula and ordinary vector addition.[2]

In this paper, we present a formalization of gyrovector spaces and their connections with hyperbolic geometry and special relativity, within the interactive theorem prover Isabelle/HOL.¹ Machine-verified proofs are gaining popularity, and with recent advances in AI, it may only be a matter of time before they become the norm. Such proofs are particularly valuable in areas prone to error; we argue that non-associative and non-commutative algebras belong to this category. Namely, human mathematicians, accustomed to the standard laws of groups and vector spaces, may make assumptions in pen-and-paper proofs that seem valid to those not specifically trained to think within non-commutative and non-associative structures. Additionally, since most gyrogroup and gyrovector space laws are expressed as equalities, automated provers, especially those based on term rewriting, often find proofs that are shorter

¹The full formalization is available at <https://github.com/jecamark96/moebius>

and more elegant than those produced by humans. We support this claim by providing several examples.

We follow the book [2], while filling in many important details that are omitted in the original text. This includes making the underlying notation more precise and providing many proofs missing from the book.

The paper is structured as follows. First, we describe the formalization of gyrogroups (Section 3) and gyrovector spaces (Section 4), defining these abstract structures and proving their essential properties. In Section 5, we formally demonstrate that Möbius transformations, traditionally used to define hyperbolic geometry, give rise to gyrogroup and gyrovector space structures (formalizing the so-called Möbius gyrovector spaces). We also formally prove that Einstein’s velocity addition follows the same structure (formalizing the so-called Einstein gyrovector spaces). In Section 6, we employ gyrovector spaces to define hyperbolic geometry in a syntactically elegant manner. We then prove that these definitions yield models equivalent to the Poincaré and Klein-Beltrami disks. Finally, in Section 7, we present our conclusions and propose potential directions for future work.

2 Related work

Over the past decade, significant progress has been made in formalizing various geometries and their models using proof assistants.

GeoCoq², developed by Narboux et al., is a formalization of geometry using the Coq proof assistant. It contains a formalization of a systematic development of Euclidean geometry based on Tarski’s axioms [3, 4]. Magaud, Narboux and Schreck investigated how projective plane geometry can be formalized in a proof assistant [5]. Narboux, Boutry and Braun used a proof assistant Coq to formally prove that Tarski’s axioms for plane neutral geometry can be derived from the corresponding Hilbert’s axioms[6]. They have also mechanized the proof that Tarski’s version of the parallel postulate is equivalent to the Playfair’s postulate used by Hilbert[7] and proved that Hilbert’s axioms for plane neutral geometry (excluding continuity) can be derived from the corresponding Tarski’s axioms[8].

Petrović and Marić describe formalization of analytic geometry of the Cartesian plane in Isabelle/HOL [9]. Boutry, Braun and Narboux describe the formalization of the arithmetization of Euclidean plane geometry in the Coq proof assistant, developing analytic geometry of the Cartesian within the Tarski’s axiom system[10].

Makarios formalized Klein/Beltrami disc model of hyperbolic geometry in the projective space $\mathbb{R}P^2$ in Isabelle/HOL[11]. Based on the work of Makarios, Harrison has shown the independence of Euclid’s axiom in HOL-Light, and Coghetto formalized the Klein-Beltrami model within Mizar[12, 13]. Marić and Simić formally addressed the geometry of complex numbers and presented a fully mechanically verified development within the theorem prover Isabelle/HOL[14]. Simić, Marić and Boutry developed a formalization of the Poincaré disc model of hyperbolic geometry within the Isabelle/HOL proof assistant[15]. They showed that model is defined within the complex projective line $\mathbb{C}P^1$ and adheres to Tarski’s axioms, except for Euclid’s axiom, which it negates

²<https://geocoq.github.io/GeoCoq/>

while satisfying the existence of limiting parallels. The formalization follows Schwerdfeger’s book[1] based on linear algebra (points are represented by vectors of complex homogenous coordinates, circles and lines are represented by Hermitean matrices, and Möbius transformations by non-degenerate matrices). Our paper builds upon this work, leveraging some of the established results. This connection is elaborated further in Section 6.

There have been attempts to describe some fragments of physics with the help of mathematical logic. A very interesting paper about giving the axioms of relativistic theory is [16], but it lacks a computer formalization, making it a potential path for future work. Another theoretical paper about the axiomatization of physics is A. Kapustin’s paper[17]. Regarding formalization, we are familiar with Jacques Fleuriot’s paper involving both geometry and physics[18]. As far as we know, this is the first attempt to formalize gyrovector spaces.

2.1 Background

In this section we will describe Isabelle/HOL³ and its main features that are used for our formalization.

Isabelle is a generic proof assistant that supports various logics, with the most popular being HOL (Higher-Order Logic). Users specify mathematical theories, defines new concepts, states their properties and provides proofs written in a specialized, declarative proof language called Isar[19]. These proofs are checked by the system, guaranteeing their correctness.

Algebraic structures are formalized using *locales*[20] and *classes*[21]. In Isabelle/HOL, *locales* are particularly useful for defining and reasoning about algebraic structures such as groups, rings, and vector spaces, where common assumptions (like associativity, identity elements, and distributivity) are essential across different proofs. A locale allows users to define a set of operations and assumptions that characterize an algebraic structure and then apply these assumptions across multiple contexts without repeating foundational setup. Within a locale, the **fixes** keyword is used to declare parameters, like specific operations or elements (e.g., a binary operation \oplus and an identity element e in a group). The **assumes** keyword is used to specify the properties or axioms these parameters must satisfy (e.g., associativity of \oplus and the identity property of e). The **interpretation** command is used to instantiate a locale with specific parameters and assumptions (for example to show that integers form a group when neutral e is interpreted by 0 and the group operation \oplus is interpreted by integer addition $+$), effectively importing theorems and definitions from the locale into a new context. This allows the user to apply the abstract properties and results of an algebraic structure to concrete examples, enabling seamless reuse of previously proven theorems under the given interpretation.

Algebraic structures that have just a single operation are specified using *type classes* (very similar to Haskell) [21]. Classes are very similar to locales but the main difference is that classes are suitable for defining type-based properties and structures, while locales are more general and flexible for organizing proofs and theories with shared assumptions and parameters.

³<https://isabelle.in.tum.de/>

The *lifting/transfer package* [22] is used to transfer definitions and theorems from the raw type to the abstract type. In Isabelle/HOL, an abstract type is a type defined in terms of properties and behaviors rather than concrete representation. Examples of abstract types include subset types, quotient types, etc. Subset types are defined using **typedef** construction. For example, the set of points in the Poincaré disc can be defined as a subtype of complex numbers, containing the elements whose norm is less than 1. Two functions are automatically generated (assuming the concrete type is called B and the abstract type is called A): Rep_A and Abs_A . The role of Rep_A is to map an element of the new, abstract type to its representation in the underlying type (for example, $Rep_Poincare$ maps a point of the Poincaré disc into its representing complex number). Abs_A is essentially the inverse function of Rep_A whose role is to map an element of the underlying type to an element of the new, abstract type, provided that the element satisfies the predicate defining the subset (for example, $Abs_Poincare$ maps a complex number with norm less than one into a point of the Poincaré disc).

Functions on the abstract type can be defined by first defining functions on the underlying representation type, and then lifting those definitions to the abstract type, by means of lifting/transfer package [22]. Such definition use the keyword **lift_definition**. For example, functions that operate on points in the Poincaré disc can be defined by defining functions that operate on complex numbers, and then lifting them to the subtype. Proofs about such functions can be done by transferring them from the abstract type to the representation type.

3 Gyrogroups

In this section we give a definition of a gyrogroup, a nonassociative group-like structure that was discovered by Ungar [2]. In the same way that groups serve as a bridge between associative algebra and Euclidean geometry, gyrogroups serve as a bridge between nonassociative algebra and hyperbolic geometry and relativistic physics. They are the foundation on which analytic hyperbolic geometry rests.

Definition 1 (Groupoid). *A structure (G, \oplus) where \oplus is a binary operation is a groupoid if G is a non-empty set closed for \oplus .*

Definition 2 (Automorphism). *An automorphism of a groupoid (G, \oplus) is a bijective self-map of S , $\phi : G \rightarrow G$, which preserves its groupoid operation, that is, $\phi(a \oplus b) = \phi(a) \oplus \phi(b)$ for all $a, b \in G$.*

Definition 3 (Gyrogroups [2], Definition 2.5). *A groupoid (G, \oplus) is a gyrogroup if its binary operation satisfies the following axioms.*

- (G1) *There is at least one element in G called a left identity (0) that satisfies $0 \oplus a = a$ for all $a \in G$.*
- (G2) *For each element $a \in G$ there exists an element $\ominus a \in G$ called a left inverse such that $\ominus a \oplus a = 0$.*
- (G3) *For any $a, b, c \in G$ there exists a unique element $gyr[a, b]c$ from G such that a left gyroassociative law is obeyed. It means that:*

$$a \oplus (b \oplus c) = (a \oplus b) \oplus gyr[a, b]c$$

(G4) The map $\text{gyr}[a, b] : G \rightarrow G$ given by $c \mapsto \text{gyr}[a, b]c$ is an automorphism of the groupoid (G, \oplus) , that is,

$$\text{gyr}[a, b] \in \text{Aut}(G, \oplus)$$

and the automorphism $\text{gyr}[a, b]$ of G is called the gyroautomorphism, or the gyration, of G generated by $a, b \in G$. The operator $\text{gyr} : G \times G \rightarrow \text{Aut}(G, \oplus)$ is called the gyrator of G .

(G5) Finally, the gyroautomorphism $\text{gyr}[a, b]$ generated by any $a, b \in G$ possesses the left reduction property:

$$\text{gyr}[a, b] = \text{gyr}[a \oplus b, b]$$

From these axioms, it can be easily shown that there is a unique identity 0 (both left and right) and that every element $a \in G$ has its own unique inverse (both left and right). The structure of a gyrogroup is a very rich structure, and many lemmas and theorems follow from these axioms. We give example of one simple (but not trivial) lemma with a proof.

Lemma 1. (General left cancellation law): Let (G, \oplus) be a gyrogroup. For any elements $a, b, c \in G$ we have: If $a \oplus b = a \oplus c$ then $b = c$.

Proof. From (G2) we know that there must exist an element x such that $x \oplus a = 0$ where 0 is a left identity. From (G3) we conclude that $x \oplus (a \oplus b) = (x \oplus a) \oplus \text{gyr}[x, a]b$ which is also equal to $\text{gyr}[x, a]b$ since 0 is a left identity. On the other hand, $x \oplus (a \oplus b) = x \oplus (a \oplus c)$ because $a \oplus b = a \oplus c$ is a condition of this lemma. Moreover, from (G3) we also know that $x \oplus (a \oplus c) = (x \oplus a) \oplus \text{gyr}[x, a]c = \text{gyr}[x, a]c$. So, finally, we have that $\text{gyr}[x, a]b = \text{gyr}[x, a]c$. From this equation the result follows very easy, since gyroautomorphisms are bijective functions. \square

3.1 Isabelle/HOL formalization

Formalization of gyrogroups in Isabelle/HOL is very straightforward. Since it is an algebraic operation on a single type we define it as a type class[21]. First we define a groupoid and a groupoid automorphism.

```
class gyrogroupoid =
  fixes gyrozero :: "'a" ("0_g")
  fixes gyroplus :: "'a ⇒ 'a ⇒ 'a" (infixl "⊕" 100)
begin
definition gyroaut :: "('a ⇒ 'a) ⇒ bool where
  "gyroaut f ⟷ (∀ a b. f (a ⊕ b) = f a ⊕ f b) ∧ bij f"
end
```

Next we define a gyrogroup. We define subtraction using addition and left inverse.

```
class gyrogroup = gyrogroupoid +
  fixes gyroinv :: "'a ⇒ 'a" ("⊖")
  fixes gyr :: "'a ⇒ 'a ⇒ 'a ⇒ 'a"
  assumes gyro_left_id: "⋀ a. (0_g ⊕ a = a)"
  assumes gyro_left_inv: "⊖ a ⊕ a = 0_g"
  assumes gyro_left_assoc:
```

```

    "\(\ a b z. a \oplus (b \oplus z) = (a \oplus b) \oplus (gyr a b z)"
    assumes gyr_gyroaut: "\(\ a b. gyroaut (gyr a b)"
    assumes gyr_left_loop: "\(\ a b. gyr a b = gyr (a \oplus b) b"
begin
definition gyrominus :: "'a \Rightarrow 'a \Rightarrow 'a" (infixl "\ominus_b" 100) where
  "a \ominus_b b = a \oplus (\ominus b)"
end

```

Our particular interest is in gyrocommutative gyrogroups, since, as we will see, some of them give a rise to gyrovector spaces.

```

class gyrocommutative_gyrogroup = gyrogroup +
  assumes gyro_commute: "a \oplus b = gyr a b (b \oplus a)"

```

Lemma 1 can be formulated and proved in the Isar language as follows:

```

lemma gyro_left_cancel:
  assumes "a \oplus b = a \oplus c"
  shows "b = c"
proof-
  from assms have "(\ominus a) \oplus (a \oplus b) = (\ominus a) \oplus (a \oplus c)" by simp
  then have "(\ominus a \oplus a) \oplus gyr (\ominus a) a b = (\ominus a \oplus a) \oplus gyr (\ominus a) a c"
    using gyro_left_assoc by simp
  then have "gyr (\ominus a) a b = gyr (\ominus a) a c" by simp
  then show "b = c" using gyr_inj by blast

```

qed

However, good support for automated theorem proving in Isabelle/HOL can provide a much shorter proof than the one given in the book. Namely, the prover *metis* shows the lemma fully automatically, given a list of axioms and previous lemmas used in that automated proof. This proof (including the relevant list of lemmas) is found by the tool Sledgehammer[23].

```

lemma gyro_left_cancel:
  assumes "a \oplus b = a \oplus c"
  shows "b = c"

```

```

using assms
by (metis gyr_inj gyro_left_assoc gyro_left_id gyro_left_inv)

```

We have formally proved a large number of lemmas and theorems about gyrogroups (given in Chapter 2 and 3 in [2]). Our experience shows that the support for automated theorem proving in Isabelle/HOL, especially in the domain of equational reasoning that lies in the core of the proofs of gyrogroup properties, and the Sledgehammer tool for proof finding are very powerful, so we managed to prove many theorems that have long elaborate proofs in the book fully automatically (of course, such proofs do not give understanding why the statement holds).

4 Gyrovector spaces

Gyrogroups are used to define gyrovectors and gyrovector spaces.

Definition 4 (Real Inner Product Gyrovector Spaces, Definition 6.2 in [2]). *A real inner product gyrovector space (G, \oplus, \otimes) (gyrovector space, in short) is a gyrocommutative gyrogroup (G, \oplus) that obeys the following axioms:*

- (1) G is a subset of a real inner product vector space V called the carrier of G , $G \subseteq V$, from which it inherits its inner product, \cdot , and norm, $\|\cdot\|$ which are invariant under gyroautomorphisms, that is,

$$\text{gyr}[u, v]a \cdot \text{gyr}[u, v]b = a \cdot b$$

for all points $a, b, u, v \in G$.

- (2) G admits a scalar multiplication, \otimes , possessing the following properties. For all real numbers $r, r_1, r_2 \in \mathbb{R}$ and all points $a \in G$:

$$(V1) \quad 1 \otimes a = a$$

$$(V2) \quad (r_1 + r_2) \otimes a = (r_1 \otimes a) \oplus (r_2 \otimes a) \quad \text{Scalar Distributive Law}$$

$$(V3) \quad (r_1 r_2) \otimes a = r_1 \otimes (r_2 \otimes a) \quad \text{Scalar Associative Law}$$

$$(V4) \quad \frac{|r| \otimes a}{\|r \otimes a\|} = \frac{a}{\|a\|} \quad \text{Scaling Property}$$

$$(V5) \quad \text{gyr}[u, v](r \otimes a) = r \otimes \text{gyr}[u, v]a \quad \text{Gyroautomorphism Property}$$

$$(V6) \quad \text{gyr}[r_1 \otimes v, r_2 \otimes v] = I \quad \text{Identity Automorphism}$$

- (3) Real vector space structure $(\|G\|, \oplus, \otimes)$ for the set $\|G\|$ of one-dimensional "vectors":

$$\|G\| = \{\pm \|a\| : a \in G\} \subseteq \mathbb{R}$$

with vector addition \oplus and scalar multiplication \otimes , such that for all $r \in \mathbb{R}$ and $a, b \in G$,

$$(V7) \quad \|r \otimes a\| = |r| \otimes \|a\| \quad \text{Homogeneity Property}$$

$$(V8) \quad \|a \oplus b\| \leq \|a\| \oplus \|b\| \quad \text{Gyrotriangle Inequality}$$

4.1 Isabelle/HOL formalization

The use of operations might look ambiguous in some axioms, and this must be clarified in the formalization.

The first unclear place is the axiom $V4$. It uses a division – an operation which is at first sight not present in the structure (G, \otimes, \oplus) . A closer look reveals that, since G must be a subset of some real inner product vector space V , „vectors” from G can be multiplied by scalars from the field (in this case the field is \mathbb{R}) and also by their inverses (which explains a division in axiom $V4$). Note that the denominators $\|r \otimes a\|$ and $\|a\|$ are real numbers. Therefore, $V4$ uses both the gyro scalar multiplication \otimes (of $|r| \in \mathbb{R}$ and $a \in G$) and the scalar multiplication of the inner product space (of $\frac{1}{\|r \otimes a\|} \in \mathbb{R}$ and $|r| \otimes a \in G$).

Operands of \oplus and \otimes in axioms $V7$ and $V8$ are real numbers. In these axioms these real numbers (norms of some gyrovectors) are treated as elements of G , which is in accordance to the requirement that the set $\|G\| = \{\pm \|a\| : a \in G\}$ is a vector space wrt. the operations \oplus and \otimes . Therefore, real numbers must be embedded into G , i.e., there must be exist a conversion from real numbers to elements of G . There is a very natural way to do that. Since all n -dimensional real inner product spaces are isomorphic to \mathbb{R}^n , a real number x can be converted into a vector $(x, 0, \dots, 0)$. Note that that such embedding is traditionally present in the expositions of Euclidean

geometry (the same notation $+$ and \cdot is used for addition and scalar multiplication of vectors and addition and scalar multiplication of their norms – norms are a real number and they are identified with one-dimensional vectors). Also note that in gyrovector spaces the operations \oplus and \otimes have different properties when they operate on G and on $\|G\|$ (for example, in $\|G\|$ case addition \oplus is associative and commutative).

With this in mind, we can formalize gyrovector spaces. To formalize relationship between G and its carrier V (it holds that $G \subseteq V$) we use explicit embedding function *to_carrier* that maps elements of G to their formal counterparts in V . Its inverse function (on the part of V that corresponds to G) is denoted by *of_carrier*. The function *in_domain* checks if the given element is in the part of V that corresponds to G . Norm and inner product are inherited from the carrier space V .

```

locale gyrocarrier' =
  fixes to_carrier :: "'a::gyrocommutative_gyrogroup  $\Rightarrow$  'b::real_inner"
  fixes of_carrier :: "'b  $\Rightarrow$  'a"
  fixes in_domain :: "'b  $\Rightarrow$  bool"
  assumes to_carrier:
    " $\bigwedge$  b. in_domain b  $\implies$  to_carrier (of_carrier b) = b"
  assumes of_carrier: " $\bigwedge$  a. of_carrier (to_carrier a) = a"
  assumes to_carrier_zero: "to_carrier 0g = 0"
begin
definition gyronorm :: "'a  $\Rightarrow$  real" ("⟨_⟩" [100] 100) where
  "⟨a⟩ = norm (to_carrier a)"
definition gyroinner :: "'a  $\Rightarrow$  'a  $\Rightarrow$  real" (infixl "." 100) where
  "a · b = inner (to_carrier a) (to_carrier b)"
definition norms :: "real set" where
  "norms = {x.  $\exists$  a. x = ⟨a⟩}  $\cup$  {x.  $\exists$  a. x = -⟨a⟩}"
end

```

In a very similar manner we formalize embedding of $\|G\|$ (defined as **norms** in the previous locale) into G . We require that exist a function *of_real* that converts some real numbers (norms of gyrovectors) to elements of G (gyrovectors). We also assume that *to_real* is its inverse function (on its codomain).

When giving concrete interpretations of gyrovector spaces, user must supply all embedding functions. For example, in Section 5 we shall see that G can be the abstract set of points in the Poincaré disc, V is the set of complex numbers (it is a real inner product vector space) – the Poincaré disc is identified by complex numbers that have norm less than 1. The function *to_carrier* maps a point in the Poincaré disc to its representing complex number, and *of_carrier* is its inverse (on the complex unit disc). The function *of_real* uses the natural embedding of real into complex numbers (it maps the set $\|G\|$, which is in this case the open interval $(-1, 1)$ first to complex numbers and then to points in the Poincaré disc).

```

locale gyrocarrier'' = gyrocarrier' +
  fixes of_real :: "real  $\Rightarrow$  'a"
  fixes to_real :: "'a  $\Rightarrow$  real"
  assumes to_real: " $\bigwedge$  x. x  $\in$  norms  $\implies$  to_real (of_real x) = x"

```

Next we introduce the assumption that the inner product must be invariant under gyrations.

```

class gyrocarrier = gyrocarrier'' +

```

```

assumes inner_gyroauto_invariant:
  " $\bigwedge u v a b. (\text{gyr } u v a) \cdot (\text{gyr } u v b) = a \cdot b$ "
  Finally, we define gyrovector spaces as gyrocarriers that include the gyro scalar
  multiplication  $\otimes$  and satisfy the given gyrovector space axioms.
locale gyrovector_space = gyrocarrier +
  fixes scale :: "real  $\Rightarrow$  'a  $\Rightarrow$  'a" (infixl " $\otimes$ " 105)
  assumes scale_1: " $\bigwedge a. 1 \otimes a = a$ "
  assumes scale_distrib:
    " $\bigwedge r_1 r_2 a. (r_1 + r_2) \otimes a = r_1 \otimes a \oplus r_2 \otimes a$ "
  assumes scale_assoc: " $\bigwedge r_1 r_2 a. (r_1 * r_2) \otimes a = r_1 \otimes (r_2 \otimes a)$ "
  assumes scale_prop1: " $\bigwedge r a. r \neq 0 \implies$ 
     $\text{to\_carrier } (|r| \otimes a) /_R \langle\langle r \otimes a \rangle\rangle = \text{to\_carrier } a /_R \langle\langle a \rangle\rangle$ "
  assumes gyroauto_property:
    " $\bigwedge u v r a. \text{gyr } u v (r \otimes a) = r \otimes (\text{gyr } u v a)$ "
  assumes gyroauto_id: " $\bigwedge r_1 r_2 v. \text{gyr } (r_1 \otimes v) (r_2 \otimes v) = \text{id}$ "
  assumes homogeneity:
    " $\bigwedge r a. \langle\langle r \otimes a \rangle\rangle = \text{to\_real } ((\text{abs } r) \otimes (\text{of\_real } \langle\langle a \rangle\rangle))$ "
  assumes gyrotriangle:
    " $\bigwedge a b. \langle\langle a \oplus b \rangle\rangle \leq \text{to\_real } (\text{of\_real } \langle\langle a \rangle\rangle) \oplus (\text{of\_real } \langle\langle b \rangle\rangle)$ "

```

In this abstract setup we proved a large number of lemmas and theorems that hold in gyrovector spaces (given in Chapter 8 in [2]).

4.2 Basic geometric notions

The structure of gyrovector spaces is rich enough to allow defining basic geometric notions. For $x \neq y$ it holds that z belongs to the line determined by x and y iff there exists a real number t such that $x = y \oplus t \otimes (\ominus y \oplus z)$ (which is fully analogous to the Euclidean case $x = y + t(z - y)$). This allows to introduce the definition of collinearity (note that we also include a degenerate case where $x = y$).

```

definition collinear :: "'a  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  bool" where
  "collinear x y z  $\longleftrightarrow$  ( $y = z \vee (\exists t::\text{real}. (x = y \oplus t \otimes (\ominus y \oplus z)))$ )"

```

Collinearity enables the definition of lines.

```

definition gyroline :: "'a  $\Rightarrow$  'a  $\Rightarrow$  'a set" where
  "gyroline a b = {x. collinear x a b}"

```

Definition of the betweenness relation is quite similar (and again analogous to the Euclidean case).

```

definition between :: "'a  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  bool" where
  "between x y z  $\longleftrightarrow$ 
    ( $\exists t::\text{real}. 0 \leq t \wedge t \leq 1 \wedge y = x \oplus t \otimes (\ominus x \oplus z)$ )"

```

Gyro-distance between two points is defined as the norm of their difference (again analogously to the Euclidean case).

```

definition distance :: "'a  $\Rightarrow$  'a  $\Rightarrow$  real" where
  "distance u v =  $\langle\langle \ominus u \oplus v \rangle\rangle$ "

```

It is very easy to prove basic properties of these notions. For example, one very important property is that all these properties are invariant under isometries. We define translation that maps a given point a to the origin, and prove that it preserves

basic geometric notions (proofs of these are very simple and fully automated, using previously proven lemmas about gyrovector spaces).

```

definition translate :: "'a  $\Rightarrow$  a  $\Rightarrow$  a" where
  "translate a x =  $\ominus$  a  $\oplus$  x"
lemma collinear_translate:
  shows "collinear u v w  $\longleftrightarrow$ 
    collinear (translate a u) (translate a v) (translate a w)"
lemma between_translate:
  shows "between u v w  $\longleftrightarrow$ 
    between (translate a u) (translate a v) (translate a w)"
lemma distance_translate:
  shows "distance u v = distance (translate a u) (translate a v)"

```

As an illustration, we give an informal proof that translations preserve collinearity (in Isabelle/HOL this is proved by automated provers).

Theorem 1. *Let t_a be a translation $t_a(x) = \ominus a \oplus x$. Three points x , u and v are collinear iff $t_a(x)$, $t_a(u)$ and $t_a(v)$ are collinear.*

Proof. Points x , u and v are gyrocollinear, iff $u = v$ or there exists a real number t such that $x = u \oplus t \otimes (\ominus u \oplus v)$. We can prove that this formula is equivalent to $\ominus a \oplus u = \ominus a \oplus v$ or there exists a real number t such that

$$(\ominus a \oplus x) = (\ominus a \oplus u) \oplus t \otimes (\ominus(\ominus a \oplus u) \oplus (\ominus a \oplus v)).$$

Indeed, if $x = u \oplus t \otimes (\ominus u \oplus v)$, then

$$\begin{aligned}
 (\ominus a \oplus x) &= \ominus a \oplus (u \oplus t \otimes (\ominus u \oplus v)) \\
 &= (\ominus a \oplus u) \oplus \text{gyr}[\ominus a, u](t \otimes (\ominus u \oplus v)) \\
 &= (\ominus a \oplus u) \oplus t \otimes \text{gyr}[\ominus a, u](\ominus u \oplus v) \\
 &= (\ominus a \oplus u) \oplus t \otimes (\ominus(\ominus a \oplus u) \oplus (\ominus a \oplus v))
 \end{aligned}$$

We used gyroassociativity, gyroautomorphism property and the gyrotranslation theorem that states that for a , b and c , it holds that $\ominus(a \oplus b) \oplus (a \oplus c) = \text{gyr}[a, b](\ominus b \oplus c)$ (Theorem 3.13 in [2], that we formally proved in Isabelle/HOL).

The other direction is very similar. □

5 Instances of gyrovector spaces

So far gyrogroups and gyrovector spaces have been presented as abstract algebraic structures. In this section we shall describe their two most important instances: Möbius gyrovector space coming from the realm of hyperbolic geometry and Einstein gyrovector space coming from the realm of relativistic physics. In order to define these spaces we must define the operations \oplus and \otimes .

5.1 Möbius gyrogroup

Möbius gyrovector space comes from the Poincaré disc model of hyperbolic geometry. In [15] it was formally proved that all direct isometries of the Poincaré disc (given in the complex plane) are of the form:

$$z \mapsto e^{i\theta} \frac{a + z}{1 + \bar{a}z}.$$

This inspires the following definition of addition.

Definition 5 (Möbius addition in the Poincaré disc, Definition 3.125 in [2]). *Let $V_1 = \{z \in \mathbb{C} : \|z\| < 1\}$ be a unit disc in \mathbb{C} . Möbius addition \oplus_m is a binary operation defined in V_1 as:*

$$u \oplus_m v = \frac{u + v}{1 + \bar{u}v}$$

Then $z \mapsto a \oplus_m z$ is a translation of the Poincaré disc that maps $-a$ to zero⁴, $z \mapsto e^{i\theta} z$ is a rotation of the disc for the angle θ and each direct isometry is a translation followed by a rotation.

Recall that gyration function is necessary to fix the associativity and commutativity. Therefore, in the case of Möbius gyrogroup, gyration is defined by

$$\text{gyr}_m[a, b]z = \frac{a \oplus_m b}{b \oplus_m a} z = \frac{1 + a\bar{b}}{1 + \bar{a}b} z.$$

The unit disc in \mathbb{C} (the Poincaré disc) equipped with the Möbius addition and this gyration forms a gyrocommutative gyrogroup.

Note that Möbius addition can be naturally generalized from the unit disc in \mathbb{C} to a unit ball in any finite dimension real inner product vector space V (all of those are isomorphic to \mathbb{R}^n). However, we did not consider this in our formalization.

Definition 6 (Möbius addition in the ball of V). *Let V be a real inner product vector space, let $V_s = \{v \in V : \|v\| < s\}$ be a ball in V . Möbius addition \oplus_m is a binary operation defined in V_s as:*

$$u \oplus_m v = \frac{(1 + \frac{2}{s^2}uv + \frac{1}{s^2}\|v\|^2)u + (1 - \frac{1}{s^2}\|u\|^2)v}{1 + \frac{2}{s^2}uv + \frac{1}{s^4}\|u\|^2\|v\|^2}$$

5.1.1 Isabelle/HOL formalization

In our formalization we first defined the set \mathbb{P} of points in the Poincaré disc as a subtype of complex numbers \mathbb{C} . Therefore we consider real inner product space $V = \mathbb{C}$ and the gyrogroup domain $G = \mathbb{P}$.⁵

```
typedef PoincareDisc = "{z::complex. cmod z < 1}"
```

```
setup_lifting type_definition_PoincareDisc
```

This lifting introduces the function `of_complex` and `to_complex` that converts between \mathbb{C} and \mathbb{P} .

⁴Such translations are sometimes called Blaschke factors.

⁵Note that we introduced the notation \mathbb{P} for the same set that is denoted by V_1 in [2].

Then we used the lifting/transfer package to define functions operating on the Poincaré disc \mathbb{P} by first defining them on the representation type \mathbb{C} . For example, inner product and norm are naturally lifted from \mathbb{C} to the Poincaré disc (and the notation \cdot and $\langle _ \rangle$ is introduced).

lift_definition inner_p ::

"PoincareDisc \Rightarrow PoincareDisc \Rightarrow real" (infixl "." 100) is inner

lift_definition norm_p :: "PoincareDisc \Rightarrow real" (" $\langle _ \rangle$ " [100] 101) is norm

We define the Möbius addition by the following sequence of definitions (first we define it on the representation type \mathbb{C} and then we lift it to the abstract type \mathbb{P} of points of the Poincaré disc).

definition oplus_m' :: "complex \Rightarrow complex \Rightarrow complex" where

"oplus_m' u v = (u + v) / (1 + (cnj u)*v)"

lift_definition oplus_m:: "PoincareDisc \Rightarrow PoincareDisc \Rightarrow PoincareDisc"

(infixl " \oplus_m " 100) is oplus_m'

Note that this definition must be followed by a proof that the result of the defined function always belongs to the unit disc if its arguments are in the unit disc.

lemma oplus_m'_in_disc:

fixes u v :: complex

assumes "cmod u < 1" "cmod v < 1"

shows "cmod (oplus_m' u v) < 1"

The proof of this lemma is not trivial, but is quite technical and non-interesting. Note that such proofs are usually missing from the literature (they are not even mentioned in [2]), but the process of formalization reveals all the properties that must be proved in order to have a fully formalized theory.

We formalize the gyration by the next sequence of definitions.

definition gyr_m' :: "complex \Rightarrow complex \Rightarrow complex \Rightarrow complex" where

"gyr_m' a b z = ((1 + a * cnj b) / (1 + cnj a * b)) * z"

lift_definition gyr_m ::

"PoincareDisc \Rightarrow PoincareDisc \Rightarrow PoincareDisc \Rightarrow PoincareDisc" is gyr_m'

For this definition to be valid, it must be proved that the result of gyration is in the unit disc if all its arguments are in the unit disc (which is easy, since $|1+a\bar{b}| = |1+\bar{a}b|$).

lemma gyr_m'_in_disc:

fixes a b z :: complex

assumes "cmod a < 1" "cmod b < 1" "cmod z < 1"

shows "cmod (gyr_m' a b z) < 1"

One of the central results of our formalization is that with such gyration, (\mathbb{P}, \oplus_m) is a gyrocommutative gyrogroup, i.e., that \oplus_m satisfies all axioms of gyrogroup and the axiom of gyrocommutativity (this last proof is very easy, since gyration is defined so that the gyrocommutativity holds). Again, these proofs are not present in the literature (at one point Ungar comments „this space turns out to be a gyrocommutative gyrogroup, as one can readily check by computer algebra”). They are quite technical and indeed reduce to algebraic manipulations with complex numbers, but, in order to formalize them we had to make all the computation steps explicit (they are available in our formalization, but we do not print them in the paper).

As it is often the case, it turns out that it is useful to have alternative definitions of the same notion, so we formalized (and later used) an alternative definition of Möbius addition (and, of course, proved that it is equivalent to the original one).

```

definition oplus_m'_alternative :: "complex  $\Rightarrow$  complex  $\Rightarrow$  complex" where
  "oplus_m'_alternative u v =
    ((1 + 2 * inner u v + (norm v)2) *R u + (1 - (norm u)2) *R v) /
    (1 + 2 * inner u v + (norm u)2 * (norm v)2)"
lift_definition oplus_m_alternative ::
  "PoincareDisc  $\Rightarrow$  PoincareDisc  $\Rightarrow$  PoincareDisc" is oplus_m'_alternative

```

5.2 Möbius gyrovector space

In [2] Ungar shows that Möbius gyrogroup (V_1, \oplus_m) admits scalar multiplication \otimes_m , turning it into the Möbius gyrovector space $(V_1, \oplus_m, \otimes_m)$.

Definition 7 (Möbius Scalar Multiplication). *Let (V_1, \oplus_m) be a Möbius gyrogroup. For $r \in \mathbb{R}$, $v \in V_1$, and $v \neq 0$, the Möbius scalar multiplication $r \otimes_m v$ is in V_1 and is given by the equation:*

$$r \otimes_m v = \frac{(1 + \|v\|)^r - (1 - \|v\|)^r}{(1 + \|v\|)^r + (1 - \|v\|)^r} \cdot \frac{v}{\|v\|}. \quad (1)$$

For $v = 0$, it holds that $r \otimes_m 0 = 0$.

This formula is derived by generalizing the expressions for $2 \otimes v = v \oplus v$, $3 \otimes v = v \oplus v \oplus v$ etc.

5.2.1 Isabelle/HOL formalization

As expected, first we define our \otimes_m operation on complex numbers and then we lift this definition to the Poincaré disc.⁶

```

definition otimes_me'_k :: "real  $\Rightarrow$  complex  $\Rightarrow$  real" where
  "otimes_me'_k r z = ((1 + cmod z) powr r - (1 - cmod z) powr r) /
    ((1 + cmod z) powr r + (1 - cmod z) powr r)"
definition otimes_me' :: "real  $\Rightarrow$  complex  $\Rightarrow$  complex" where
  "otimes_me' r z = (if z = 0 then 0 else otimes_me'_k r z * (z / cmod z))"
lift_definition otimes_me :: "PoincareDisc  $\Rightarrow$  PoincareDisc  $\Rightarrow$  PoincareDisc"
  (infixl " $\otimes_{me}$ " 105) is me_otimes'

```

We did not find an explicit proof that $(\mathbb{P}, \oplus_m, \otimes_{me})$ is a gyrovector space in the literature (only some fragments were presented in the books[2]). So, another contribution of our formalization is that this proof is given explicitly (and additionally, its correctness is machine checked).

Very important characterization of \otimes_{me} uses hyperbolic trigonometric functions (instead of power function used in the original definition).

```

lemma otimes_me'_k_tanh:
  fixes r :: real and z :: complex
  assumes "cmod z < 1"
  shows "otimes_me'_k r z = tanh (r * artanh (cmod z))"

```

Our proof would be much harder (or almost impossible) if we didn't use the Lorentz factor, a crucial concept in the theory of relativity, named after the Dutch physicist Hendrik Lorentz. It describes how time, length, and relativistic mass change for an object moving relative to an observer. The Lorentz factor γ_u is given by the equation:

⁶In section 5.3 we shall see that the definition of scalar multiplication is exactly the same in the Einstein gyrovector space, so we denote this operation, by \otimes_m , \otimes_e and by \otimes_{me} , depending on the context.

$$\gamma_u = \frac{1}{\sqrt{1 - \frac{\|u\|^2}{c^2}}}$$

where c is the speed of the light in vacuum.

A special property of Lorentz factor is that Lorentz factor is complex if the norm of u is bigger than 1 and real if the norm is smaller than 1 and vice versa. However, we could not use this in Isabelle/HOL since functions which define power (`sqrt`, `pow`, etc.) are always real valued.

We defined the Lorentz γ -factor in the following way (assuming without loss of generality that $c = 1$):

```

definition gamma_factor :: "complex  $\Rightarrow$  real" (" $\gamma$ ") where
  "gamma_factor u =
    (if (norm u)2 < 1 then 1 / sqrt (1 - (norm u)2) else 0)"
lift_definition gamma_factor_p :: "PoincareDisc  $\Rightarrow$  real" (" $\gamma_p$ ") is
  gamma_factor

```

Most gyrovector space axioms were non-trivial, but rather straightforward to prove, but there were exceptions. The hardest axiom to prove was the gyrotriangle inequality. We needed to prove that $\|a \oplus_m b\| \leq \|a\| \oplus_m \|b\|$ and we did that using Lorentz factors and their properties.

First we proved a lemma that claims that γ -factor is a monotonically increasing function on the interval $[0, 1)$ and so its inverse function.

```

lemma gamma_factor_increase_reverse:
  fixes t1 t2 :: real
  assumes "0  $\leq$  t1" "t1 < 1" "0  $\leq$  t2" "t2 < 1"
  assumes " $\gamma$  t1 >  $\gamma$  t2"
  shows "t1 > t2"

```

After that we have proved the following lemma that expresses the γ -factor of the norm of the Möbius sum of two points in terms of their γ -factors.

```

lemma gamma_factor_norm_oplus_m:
  fixes a b :: PoincareDisc
  shows " $\gamma$   $\langle\langle$  a  $\oplus_m$  b  $\rangle\rangle$  =
     $\gamma_p$  a *  $\gamma_p$  b * cmod (1 + cnj (to_complex a) * (to_complex b))"

```

From this, we could prove the triangle inequality for the γ -factors.

```

lemma gamma_factor_oplus_m_triangle_inequality:
  fixes a b :: PoincareDisc
  shows " $\gamma$   $\langle\langle$  a  $\oplus_m$  b  $\rangle\rangle$   $\leq$   $\gamma_p$  ((of_complex  $\langle\langle$  a  $\rangle\rangle$ )  $\oplus_m$  (of_complex  $\langle\langle$  b  $\rangle\rangle$ ))"

```

The triangle inequality holds from this by the monotonicity of the inverse γ -factor.

Proof of these and other related lemmas are available in our formalization. They are non-trivial and rely on algebraic manipulations of expressions (equalities and inequalities) over complex numbers.

5.3 Einstein gyrogroup and gyrovector space

According to the Einstein's theory of relativity, as an object's speed approaches the speed of light, its relativistic mass increases towards infinity, requiring infinite energy to accelerate further. Therefore, it is impossible for an object to move at or above the speed of light. While Newtonian velocities are represented by vectors in \mathbb{R}^3 , Einstein

velocities are represented by vectors within a ball of radius c where c is the speed of light. In our formalization we represented velocities by points of a unit ball (setting $c = 1$) in 2d space (unit disc). Most of the theorems that we formally proved in this special case can be easily generalized to higher dimensions and different values of the constant c (the speed of light, i.e., the radius of the admissible velocity ball).

In classic Newtownian physics, velocities are combined by the vector addition in an ordinary vector space. Einstein's addition law, which describes how velocities combine in special relativity, leads to rich nonassociative algebraic structures — gyrovector spaces. We formally define Einstein's velocity addition law, the corresponding Thomas gyration, and the Einstein scalar multiplication, and then we proved that they satisfy the axioms of gyrovector space.

As we shall see, it turns out that the Einstein gyrovector spaces are isomorphic to the Möbius gyrovector spaces (described in previous section), and that relativistic physics is set in hyperbolic geometry. We formalized this connection between Möbius and Einstein gyrovector spaces. As we know, this is the first explicit proof of this theorem in the literature. It is not hard, but it's technically very challenging and involves many computational tricks.

Let us first define Einstein's velocity addition \oplus_e .

Definition 8 (Einstein addition in the ball, Definition 3.39 in [2]). *Let V be a real inner product vector space and let $V_1 = \{v \in V : \|v\| < 1\}$ be a unit ball in V . Einstein addition \oplus_e is a binary operation defined in V as:*

$$u \oplus_e v = \frac{1}{1 + u \cdot v} \left\{ u + \frac{1}{\gamma_u} v + \frac{\gamma_u}{1 + \gamma_u} (u \cdot v) u \right\}$$

where γ_u is the Lorentz γ -factor and \cdot and $\|\cdot\|$ are the inner product and norm that the ball V_1 inherits from its space V .

Einstein scalar multiplication \otimes_e is defined in the same way as in the Möbius case (Equation 1).

5.3.1 Isabelle/HOL formalization

Einstein addition in \mathbb{P} is formally defined by the following definitions.

definition `oplus_e' :: "complex \Rightarrow complex \Rightarrow complex" where`

```
"oplus_e' u v =
  (1 / (1 + inner u v)) *R
  (u + (1 /  $\gamma$  u) *R v + (( $\gamma$  u / (1 +  $\gamma$  u)) * (inner u v)) *R u)"
```

lift_definition `oplus_e :: "PoincareDisc \Rightarrow PoincareDisc \Rightarrow PoincareDisc"`

```
(infixl " $\oplus_e$ " 100) is oplus_e'
```

To lift this definition from complex numbers to Poincaré disc, one must prove that the Poincaré disc is closed under this operation.

lemma `oplus_e'_in_unit_disc:`

```
assumes "cmod u < 1" "cmod v < 1"
shows "cmod (oplus_e' u v) < 1"
```

Given the complexity of the expression in the definition of Einstein addition, this proof is not trivial. We managed to prove it using γ -factors. The following lemma expresses the γ -factor of the Einstein sum of two points in terms of their γ -factors.

Lemma 2. *For all points u and v in the unit disc \mathbb{P} it holds:*

$$\gamma_{u \oplus_e v} = \gamma_u \gamma_v (1 + u \cdot v)$$

It is formalized as follows (the formalization gives the statement on the representation type \mathbb{C}):

```
lemma gamma_factor_oplus_e':
  assumes "cmod u < 1" "cmod v < 1"
  shows "γ (oplus_e' u v) = (γ u) * (γ v) * (1 + inner u v)"
```

The proof of this lemma is not hard, but is technically involved. With this lemma it is not hard to prove that the unit disc is closed for `oplus_e`. Since $\gamma (\text{oplus_e}' u v) = 1 / \sqrt{1 - (\text{cmod} (\text{oplus_e}' u v))^2}$, and $(\gamma u) * (\gamma v) * (1 + \text{inner } u v)$ is a positive real number, so must be $\sqrt{1 - (\text{cmod} (\text{oplus_e}' u v))^2}$. This is possible only if the radicand is non-negative, implying that $\text{cmod} (\text{oplus_e}' u v)$ is less than 1.

Again, details of such proofs are not present in the literature. Ungar just gives a brief comment: „Like Möbius addition in the ball, one can show by computer algebra that Einstein addition in the ball is a gyrocommutative gyrogroup operation”. This would be very challenging to do in an interactive theorem prover, so we decided to take a different approach.

5.4 Gyrovector space isomorphism

Instead of directly proving that the Einstein space $(V_1, \oplus_e, \otimes_e)$ is a gyrovector space, for which we think would be very technically challenging to do in an interactive theorem prover, we showed that this structure is isomorphic to Möbius gyrovector space $(V_1, \oplus_m, \otimes_m)$, and therefore must satisfy axioms of gyrovector space.

We begin with the following definition.

Definition 9. *Two gyrovector spaces $(G_1, \oplus_1, \otimes_1)$ and $(G_2, \oplus_2, \otimes_2)$ are isomorphic if there exists a bijective map $\phi : G_1 \rightarrow G_2$ such that the following conditions are satisfied:*

- $\phi(u \oplus_1 v) = \phi(u) \oplus_2 \phi(v)$
- $\phi(r \otimes_1 v) = r \otimes_2 \phi(v)$
- $\frac{\phi(u)}{\|\phi(u)\|} \cdot \frac{\phi(v)}{\|\phi(v)\|} = \frac{u}{\|u\|} \cdot \frac{v}{\|v\|}$

Informally, it means that ϕ preserves gyrovector space operations and keeps the inner product of the unit vectors invariant.

This definition proved to be very useful, because if we can formally prove that some structure is a gyrovector space, and we find an isomorphism map to another structure, we can easily prove that that other is also a gyrovector space.

In our formalization we have formally proved that the map $u \mapsto \frac{1}{2} \otimes_m u$ is an isomorphism that maps Möbius gyrovectors to Einstein gyrovectors, and used that to prove that Einstein gyrovectors satisfy all gyrovector space axioms. The description of this isomorphism is given in [2], but as in many other cases the proof is missing. We made a proof and formalized it in Isabelle/HOL. As a contribution to the literature we give an informal description of this proof.

First we formulate some simple lemmas.

Lemma 3. *For any point $u \in \mathbb{P}$ it holds:*

$$\frac{1}{2} \otimes_m u = \frac{\gamma_u}{\gamma_u + 1} u$$

Lemma 4. *For any point $u \in \mathbb{P}$ it holds:*

$$\left\| \frac{\gamma_u}{\gamma_u + 1} u \right\|^2 = \frac{\gamma_u^2}{(1 + \gamma_u)^2} \cdot \|u\|^2 = \frac{\gamma_u - 1}{\gamma_u + 1}$$

Our central theorem proves that the isomorphism $u \mapsto \frac{1}{2} \otimes_m u$ connects Möbius and Einstein addition (other two properties are much easier to prove).

Theorem 2. *Let $G_e = (V_c, \oplus_e, \otimes_e)$ and $G_m = (V_c, \oplus_m, \otimes_m)$ be respectively, the Einstein and the Möbius gyrovector spaces of the same ball V_c of a same real inner product space V . The following formulas are correct:*

$$\frac{1}{2} \otimes_e (u \oplus_e v) = \left(\frac{1}{2} \otimes_m u \right) \oplus_m \left(\frac{1}{2} \otimes_m v \right)$$

Proof. By Lemma 3 it holds that

$$\left(\frac{1}{2} \otimes_m u \right) \oplus_m \left(\frac{1}{2} \otimes_m v \right) = \left(\frac{\gamma_u}{\gamma_u + 1} u \right) \oplus_m \left(\frac{\gamma_v}{\gamma_v + 1} v \right)$$

It turns out that it is better to apply the alternative definition of Möbius addition (described in Section 5.1). Then we get that the right hand side is equal to:

$$\frac{\left(1 + 2 \cdot \frac{\gamma_u}{\gamma_u + 1} \cdot \frac{\gamma_v}{\gamma_v + 1} (u \cdot v) + \left\| \frac{\gamma_v}{\gamma_v + 1} v \right\|^2 \right) \cdot \frac{\gamma_u}{\gamma_u + 1} u + \left(1 - \left\| \frac{\gamma_u}{\gamma_u + 1} u \right\|^2 \right) \cdot \frac{\gamma_v}{\gamma_v + 1} v}{1 + 2 \frac{\gamma_u}{\gamma_u + 1} u \cdot \frac{\gamma_v}{\gamma_v + 1} v + \left\| \frac{\gamma_u}{\gamma_u + 1} u \right\|^2 \cdot \left\| \frac{\gamma_v}{\gamma_v + 1} v \right\|^2}$$

By applying Lemma 4, the expression transforms into:

$$\frac{\left(1 + \frac{2\gamma_u\gamma_v}{(\gamma_u+1)(\gamma_v+1)} (u \cdot v) + \frac{\gamma_v-1}{\gamma_v+1} \right) \cdot \frac{\gamma_u}{\gamma_u+1} u + \left(1 - \frac{\gamma_u-1}{\gamma_u+1} \right) \cdot \frac{\gamma_v}{\gamma_v+1} v}{1 + \frac{2\gamma_u\gamma_v}{(\gamma_u+1)(\gamma_v+1)} (u \cdot v) + \frac{\gamma_u-1}{\gamma_u+1} \cdot \frac{\gamma_v-1}{\gamma_v+1}}$$

We simplify our expression more using the following equations:

$$1 + \frac{\gamma_v - 1}{\gamma_v + 1} = \frac{2\gamma_v}{\gamma_v + 1}, \quad 1 - \frac{\gamma_u - 1}{\gamma_u + 1} = \frac{2}{\gamma_u + 1}$$

$$1 + \frac{\gamma_u - 1}{\gamma_u + 1} \cdot \frac{\gamma_v - 1}{\gamma_v + 1} = \frac{2(1 + \gamma_u\gamma_v)}{(1 + \gamma_u)(1 + \gamma_v)}$$

and we get:

$$\frac{\left(\frac{2\gamma_v}{1+\gamma_v} + \frac{2\gamma_u\gamma_v}{(1+\gamma_u)(1+\gamma_v)} (u \cdot v) \right) \cdot \frac{\gamma_u}{1+\gamma_u} u + \frac{2}{1+\gamma_u} \cdot \frac{\gamma_v}{1+\gamma_v} v}{\frac{2\gamma_u\gamma_v}{(\gamma_u+1)(\gamma_v+1)} (u \cdot v) + \frac{2(1+\gamma_u\gamma_v)}{(1+\gamma_u)(1+\gamma_v)}}$$

Adding fractions after reducing them to their common denominator, and cancelling $(1 + \gamma_u)(1 + \gamma_v)$ (that is not zero) gives:

$$\frac{2\gamma_u\gamma_v\left(u + \frac{1}{\gamma_u}v + \frac{\gamma_u}{1+\gamma_u}(u \cdot v)u\right)}{2\gamma_u\gamma_v(1 + u \cdot v) + 2} = \frac{\gamma_u\gamma_v\left(u + \frac{1}{\gamma_u}v + \frac{\gamma_u}{1+\gamma_u}(u \cdot v)u\right)}{\gamma_u\gamma_v(1 + u \cdot v) + 1}$$

On the other hand, from Lemma 3 it holds that:

$$\frac{1}{2} \otimes_e (u \otimes_e v) = \frac{\gamma_{u \otimes_e v}}{\gamma_{u \otimes_e v} + 1} (u \otimes_e v).$$

From Lemma 2 and Definition 8 of Einstein addition, we can conclude that:

$$\frac{1}{2} \otimes_e (u \otimes_e v) = \frac{\gamma_u\gamma_v(1 + u \cdot v)}{\gamma_u\gamma_v(1 + u \cdot v) + 1} \left\{ \frac{1}{1 + u \cdot v} \left(u + \frac{1}{\gamma_u}v + \frac{\gamma_u}{\gamma_u + 1}(u \cdot v) \cdot u \right) \right\}$$

Cancelling $1 + u \cdot v$ (that is not zero), and putting the two sides together, we see that the theorem holds. \square

Formal statement of this theorem is quite direct.

theorem iso_me_oplus:

shows " $(1/2) \otimes_{me} (u \oplus_e v) = ((1/2) \otimes_{me} u) \oplus_m ((1/2) \otimes_{me} v)$ "

Its formal proof directly follows the informal one, carefully checking every step of computation and simplification.

The previous result enables expressing Einstein addition in terms of Möbius addition and scalar multiplication, which are operations whose properties are by now formally analyzed.

lemma oplus_e_oplus_m:

shows " $u \oplus_e v = 2 \otimes_{me} (((1/2) \otimes_{me} u) \oplus_m ((1/2) \otimes_{me} v))$ "

Since many axioms use gyrations we needed to prove that the isomorphism map $u \mapsto (1/2) \otimes_m u$ preserves gyrations, which enables expressing Einstein gyration via Möbius gyration.

lemma iso_me_gyr:

shows " $(1/2) \otimes_{me} \text{gyr}_e a b z = \text{gyr}_m ((1/2) \otimes_{me} a) ((1/2) \otimes_E b) ((1/2) \otimes_{me} z)$ "

lemma gyr_e_gyr_m:

shows " $\text{gyr}_e a b z = 2 \otimes_{me} (\text{gyr}_m ((1/2) \otimes_{me} a) ((1/2) \otimes_E b) ((1/2) \otimes_{me} z))$ "

From these connections it is very easy to prove that Einstein addition and gyration form a gyrocommutative gyrogroup (without unfolding the definition of \oplus_E dealing with lots of hard calculations). Here is an example of one such simple and readable proof (in Isar language). The lemma proves that Einstein addition is gyroassociative.

lemma oplus_e_gyro_left_assoc:

shows " $a \oplus_e (b \oplus_e z) = (a \oplus_e b) \oplus_e \text{gyr}_e a b z$ "

proof-

let ?a = " $(1/2) \otimes_{me} a$ " **and** ?b = " $(1/2) \otimes_{me} b$ " **and** ?z = " $(1/2) \otimes_{me} z$ "

have " $a \oplus_e (b \oplus_e z) = 2 \otimes_{me} (?a \oplus_m (?b \oplus_m ?z))$ "

using iso_me_oplus oplus_e_oplus_m **by** simp

```

also have "... = 2  $\otimes_{me}$  ((?a  $\oplus_m$  ?b)  $\oplus_m$  m_gyr ?a ?b ?z)"
  using oplus_m_gyro_left_assoc by simp
also have "... = 2  $\otimes_{me}$  (((1/2)  $\otimes_{me}$  (a  $\oplus_e$  b))  $\oplus_m$  m_gyr ?a ?b ?z)"
  using iso_me_oplus by simp
also have "... = 2  $\otimes_{me}$  (((1/2)  $\otimes_{me}$  (a  $\oplus_e$  b))  $\oplus_m$  (1/2)  $\otimes_{me}$  e_gyr a b z)"
  using gyr_e_gyr_m by simp
finally show ?thesis
  using oplus_e_oplus_m by simp
qed

```

Isabelle/HOL automation is powerfull enough to find this proof fully automatically.

lemma e_gyro_left_assoc:

shows "a \oplus_e (b \oplus_e z) = (a \oplus_e b) \oplus_e e_gyr a b z"

using gyr_e_gyr_m oplus_e_oplus_m iso_me_oplus oplus_m_gyro_left_assoc
by simp

To prove gyrovector space axioms, some additional work needs to be done. We proved the connection of the isomorphism function $z \mapsto (1/2) \otimes z$ and its inverse $z \mapsto 2 \otimes z$ with the inner product and the norm. This is then used to prove the invariance of the inner product under gyrations and the triangle inequality. Since these explicit proofs are missing from [2], as a contribution to the literature we shall present an informal proof sketches (the full formal proofs are available in our formalization).

Lemma 5. *For all elements a and b in the unit disc \mathbb{P} it holds*

$$\begin{aligned}
((1/2) \otimes a) \cdot b &= \frac{\gamma_a}{1 + \gamma_a} (a \cdot b), & a \cdot ((1/2) \otimes b) &= \frac{\gamma_b}{1 + \gamma_b} (a \cdot b). \\
(2 \otimes a) \cdot b &= \frac{2\gamma_a^2}{2\gamma_a^2 - 1} (a \cdot b), & a \cdot (2 \otimes b) &= \frac{2\gamma_b^2}{2\gamma_b^2 - 1} (a \cdot b).
\end{aligned}$$

This is easily proved by direct computation. By expressing norm as $(\langle\langle a \rangle\rangle)^2 = a \cdot a$, expressions for $\langle\langle (1/2) \otimes a \rangle\rangle$ and $\langle\langle 2 \otimes a \rangle\rangle$ directly follow.

Theorem 3. *For any three points $u, v, a \in \mathbb{P}$ it holds:*

$$\text{gyr}_e[u, v]a \cdot \text{gyr}_e[u, v]b = a \cdot b.$$

Proof. Let $u' = (1/2) \otimes u$, $v' = (1/2) \otimes v$, $a' = (1/2) \otimes a$ and $b' = (1/2) \otimes b$. By the isomorphism property of gyration it holds that

$$\text{gyr}_e[u, v]a \cdot \text{gyr}_e[u, v]b = (2 \otimes \text{gyr}_m[u', v']a') \cdot (2 \otimes \text{gyr}_m[u', v']b')$$

By Lemma 5 this is further equal to:

$$\left(\frac{2\gamma_{G_a}}{2\gamma_{G_a} - 1} \right) \left(\frac{2\gamma_{G_b}}{2\gamma_{G_b} - 1} \right) (G_a \cdot G_b),$$

where $G_a = \text{gyr}_m[u', v']a'$ and $G_b = \text{gyr}_m[u', v']b'$. Since inner product is invariant under Möbius gyrations this is equal to:

$$\left(\frac{2\gamma_{G_a}}{2\gamma_{G_a} - 1} \right) \left(\frac{2\gamma_{G_b}}{2\gamma_{G_b} - 1} \right) (a' \cdot b').$$

By Lemma 5 this is further equal to:

$$\left(\frac{2\gamma_{G_a}}{2\gamma_{G_a}-1}\right)\left(\frac{2\gamma_{G_b}}{2\gamma_{G_b}-1}\right)\left(\frac{\gamma_a}{1+\gamma_a}\right)\left(\frac{\gamma_b}{1+\gamma_b}\right)(a \cdot b).$$

The proof finishes by tedious but straightforward proof that

$$\left(\frac{2\gamma_{G_a}}{2\gamma_{G_a}-1}\right)\left(\frac{\gamma_a}{1+\gamma_a}\right) = \left(\frac{2\gamma_{G_b}}{2\gamma_{G_b}-1}\right)\left(\frac{\gamma_b}{1+\gamma_b}\right) = 1$$

□

We based the proof of the gyrotriangle equality for \oplus_e on the monotonicity of the function $x \mapsto \tanh(2 \operatorname{atanh}(x))$ on the interval $[0, 1)$ and the following lemma.

Lemma 6. *For $r \in \mathbb{R}$ and $a \in \mathbb{P}$ it holds*

$$\langle\langle r \otimes a \rangle\rangle = |\tanh(r \operatorname{atanh}(\langle\langle a \rangle\rangle))|$$

For $r > 0$ (including $r = 2$) there is no need for absolute value on the right hand side.

Theorem 4. *For any $a, b \in \mathbb{P}$ it holds:*

$$\langle\langle a \oplus_e b \rangle\rangle \leq \langle\langle\langle a \rangle\rangle \oplus_e \langle\langle b \rangle\rangle\rangle$$

By the isomorphism properties left hand side is equal to:

$$\langle\langle 2 \otimes (((1/2) \otimes a) \oplus_m ((1/2) \otimes b)) \rangle\rangle$$

The right hand side is equal to:

$$\langle\langle 2 \otimes (\langle\langle (1/2) \otimes a \rangle\rangle \oplus_m \langle\langle (1/2) \otimes b \rangle\rangle) \rangle\rangle$$

The theorem then follows by the gyrotriangle inequality for Möbius addition applied on $(1/2) \otimes a$ and $(1/2) \otimes b$, Lemma 6 and the monotonicity of the function $x \mapsto \tanh(2 \operatorname{atanh}(x))$ on the interval $[0, 1)$.

Other gyrovector axioms directly follow from the isomorphism properties and were very easy to prove.

6 Modelling hyperbolic geometry

We have shown that Möbius and Einstein addition, along with gyration and scaling, give rise to gyrovector space structures. In this section, we will demonstrate that these structures correspond precisely to well-known models of hyperbolic geometry. Specifically, we will formally show that the Möbius gyrovector space is essentially equivalent to the Poincaré disc model (formally described in [15]). Since Einstein velocity addition is isomorphic to Möbius addition (as we formally established in the previous section), this implies that relativistic physics is grounded in hyperbolic geometry. It can also be shown that the Einstein gyrovector space is essentially equivalent to the Klein-Beltrami model of hyperbolic geometry (formally described in [11–13]).

To demonstrate that the Möbius gyrovector space is a valid model of hyperbolic geometry, we must show that it satisfies all axioms of absolute geometry and the negation of the parallel postulate. These axioms can be taken, for example, as Tarski's

axioms [3]. In [15], it was proven that the Poincaré disc, defined using homogeneous coordinates in projective space \mathbb{CP}^1 , satisfies these axioms. Instead of directly proving that Möbius gyrovector space concepts (such as points, collinearity, betweenness, distance, and congruence) satisfy Tarski's axioms, we have shown that they are equivalent to the corresponding concepts in the Poincaré disc model. Consequently, we know that they must satisfy all the axioms, and additionally, we know that the Möbius gyrovector space is fundamentally the same as the Poincaré disc model.

In [15], h-points in the Poincaré disc were defined as a subtype of \mathbb{CP}^1 with norm less than 1. Points in \mathbb{CP}^1 are represented by pairs of complex numbers (their homogeneous coordinates). Since all points within the Poincaré disc are finite (note that \mathbb{CP}^1 contains both finite and infinite points), these points can be identified with the complex numbers in the unit disc. Thus, we identify the unit disc \mathbb{P}' set in \mathbb{CP}^1 defined in [15] with the unit disc \mathbb{P} set in \mathbb{C} as defined in the present paper (formally, there exists a natural bijection between the two abstract types `PoincareDisc` and `PoincareDisc'`).

A fundamental aspect of each model of hyperbolic geometry is the distance function. In [15], it was shown that the distance between points in \mathbb{P}' , represented by the homogeneous coordinates of complex numbers u and v (with norm less than 1), can be calculated by:

$$d'(u, v) = \operatorname{arccosh} \left(1 + 2 \frac{\|u - v\|^2}{(1 - \|u\|^2)(1 - \|v\|^2)} \right).$$

In Section 4.2, the gyrodistance between two points in a gyrovector space is defined as $d(u, v) = \langle\langle \ominus u \oplus v \rangle\rangle$. Although gyrodistance d and distance d' are not identical, they are related by the expression $d'(u, v) = 2 \operatorname{atanh}(d(u, v))$. We have formally proved this relation (the proof involves expressing inverse trigonometric functions using logarithms, followed by calculation and expression simplification). Consequently, if we define the distance between two points in \mathbb{P} as $D(u, v) = 2 \operatorname{atanh}(\langle\langle \ominus u \oplus v \rangle\rangle)$, we obtain precisely the same distance as in the Poincaré disc model \mathbb{P}' , making the two models geometrically equivalent. Namely, since key geometric notions such as collinearity, lines, betweenness, and others can be expressed in terms of distance (for example, lines can be defined as geodesics), having an identical distance function ensures the same underlying geometry. However, rather than relying solely on this fact, we explicitly proved that betweenness and collinearity coincide in both models.

Let us demonstrate that the betweenness relations coincide (the proof for collinearity is done in a very similar way). Recall that betweenness in gyrovector spaces is defined in Section 4.2: point $x \in \mathbb{P}$ is between points $u, v \in \mathbb{P}$ (we shall this by $\mathfrak{B}(u, x, v)$) if there exists a real number $t \in [0, 1]$ such that $x = u \oplus t \otimes (\ominus u \oplus v)$. In contrast, in [15], a point $x \in \mathbb{P}'$ is between points $u, v \in \mathbb{P}'$ (we shall denote this by $\mathfrak{B}'(u, x, v)$) if and only if the cross-ratio of the points u, x, v , and the inversion of x wrt. the unit circle is a real negative number.

Theorem 5. *For any three points $u, v, x \in \mathbb{P}$ it holds that*

$$\mathfrak{B}(u, x, v) \longleftrightarrow \mathfrak{B}'(u, x, v).$$

Proof. The proof would be very challenging without using “without-loss-of-generality” reasoning. Consider the translation $z \mapsto \ominus u \oplus z$. As proved in Section 4.2, since betweenness is invariant under such translations, we have

$$\mathfrak{B}(u, x, v) \longleftrightarrow \mathfrak{B}(\ominus u \oplus u, \ominus u \oplus x, \ominus u \oplus v) \longleftrightarrow \mathfrak{B}(0, x', v'),$$

where $x' = \ominus u \oplus x$ and $v' = \ominus u \oplus v$.

On the other hand, it can simply be shown that, in the representation type of complex numbers, this translation is expressed by

$$z \mapsto \frac{z - u}{1 - \bar{u}z}.$$

In [15], it was formally demonstrated that these transformations, known as Blaschke factors, preserve many relations, including betweenness. Therefore, we have

$$\mathfrak{B}'(u, x, v) \longleftrightarrow \mathfrak{B}'(0, x', v').$$

Thus, the problem reduces to comparing the two betweenness relations in the special case when the first point is 0, i.e., proving that $\mathfrak{B}(0, x', v') \longleftrightarrow \mathfrak{B}'(0, x', v')$.

The case where $v' = 0$ is trivial, so we assume that $v' \neq 0$. From the definition of \mathfrak{B} , the left-hand side holds if and only if there exists $t \in [0, 1]$ such that $x' = t \otimes v'$. Using the characterization of \otimes , this holds if and only if

$$x' = \frac{\tanh(t \cdot \operatorname{atanh}(\|v'\|))}{\|v'\|} v'.$$

By analyzing the properties of hyperbolic trigonometric functions, we see that $t \in [0, 1]$ if and only if $\frac{\tanh(t \cdot \operatorname{atanh}(\|v'\|))}{\|v'\|} \in [0, 1]$. Therefore,

$$\mathfrak{B}(0, x', v') \longleftrightarrow (\exists t' \in [0, 1])(x' = t' \cdot v').$$

In [15], it was formally proven that this condition is equivalent to x' being between 0 and v' according to the cross-ratio-based definition of betweenness, i.e.,

$$(\exists t' \in [0, 1])(x' = t' \cdot v') \longleftrightarrow \mathfrak{B}'(0, x', v').$$

This concludes the proof. \square

Since all functions and relations from the language of Tarski’s axioms coincide between the Poincaré disc and the Möbius gyrovector space, and the Poincaré disc satisfies all of Tarski’s axioms for hyperbolic geometry, it follows that the Möbius gyrovector space does as well. Thus, although these two structures are defined very differently, they are essentially the same.

7 Conclusions and further work

In this paper, we described our Isabelle/HOL formalization of gyrogroups and gyrovector spaces [2]. These non-associative and non-commutative algebraic structures capture the essence of both hyperbolic geometry and relativistic physics. By cleverly hiding complicated computations under a few operation symbols (\oplus , \otimes), they allow

for the definition of analytic geometry in the hyperbolic plane in a syntactically almost identical way to the Cartesian geometry that models the Euclidean plane.

We have formally defined gyrogroups and gyrovector spaces, proving their numerous properties. Next, we defined two important gyrovector spaces: Möbius, inspired by hyperbolic geometry, and Einstein, inspired by relativistic physics. Proving that these concrete structures satisfy all the axioms of gyrovector spaces was challenging, as many proofs are missing in the literature (often left as “exercises for the reader” or “left to be done with computer algebra” due to their technical and computational nature). Thus, our formalization may help fill a gap in the literature about gyrovector spaces. We have formally proved that Möbius and Einstein gyrovector spaces are isomorphic. This result enabled a significantly simpler proof that Einstein’s velocity addition and related operations satisfy the axioms of gyrovector spaces, without the need to unfold the complicated definition of this operation for each axiom.

Finally, we proved that the Möbius gyrovector space is equivalent to the Poincaré disc model (previously formalized in [15]), and therefore is a model of Tarski’s axioms for hyperbolic geometry. This yields a new formal model of the Poincaré disc, with a syntactically much cleaner definition than the classical one (developed in the projective line \mathbb{CP}^1) and potentially more convenient for further research. Since the Einstein gyrovector space is isomorphic to the Möbius one, we have effectively formally proved that Einstein’s theory of relativity is set in hyperbolic geometry (a well-known, folklore, result, but to the best of our knowledge, not formalized previously).

In our future work, we plan to enrich the theory of gyrogroups and gyrovector spaces by proving additional theorems (for example we have already proved cosine theorem and Pythagorean theorem). We also plan to establish the connections between the Einstein gyrospace and the Klein-Beltrami model.

Declarations

- The authors did not receive support from any organization for the submitted work.
- All authors certify that they have no affiliations with or involvement in any organization or entity with any financial interest or non-financial interest in the subject matter or materials discussed in this manuscript.

References

- [1] Schwerdtfeger, H.: Geometry of Complex Numbers: Circle Geometry, Moebius Transformation, Non-euclidean Geometry. Courier Corporation, ??? (1979)
- [2] U., A.A.: Analytic hyperbolic geometry: Mathematical foundations and applications. (2005). <https://api.semanticscholar.org/CorpusID:118184283>
- [3] Tarski, A., Givant, S.: Tarski’s system of geometry. *Bulletin of Symbolic Logic* **5**, 175–214 (1999)
- [4] Narboux, J.: Mechanical theorem proving in tarski’s geometry, pp. 139–156 (2006). https://doi.org/10.1007/978-3-540-77356-6_9

- [5] Magaud, N., Narboux, J., Schreck, P.: Formalizing projective plane geometry in coq. *Proceedings of Automated Deduction in Geometry (ADG'08)* **6301**, 141–162 (2008)
- [6] Braun, G., Boutry, P., Narboux, J.: From hilbert to tarski. In: *Eleventh International Workshop on Automated Deduction in Geometry. Proceedings of ADG 2016*, p. 19, Strasbourg, France (2016). <https://inria.hal.science/hal-01332044>
- [7] Boutry, P., Narboux, J., Schreck, P.: Parallel postulates and decidability of intersection of lines: a mechanized study within Tarski's system of geometry. working paper or preprint (2015). <https://inria.hal.science/hal-01178236>
- [8] Braun, G., Narboux, J.: From tarski to hilbert. In: Ida, T., Fleuriot, J. (eds.) *Automated Deduction in Geometry 2012*, vol. 7993, pp. 89–109. Springer, Edinburgh, United Kingdom (2012). https://doi.org/10.1007/978-3-642-40672-0_7 . Jacques Fleuriot. <https://inria.hal.science/hal-00727117>
- [9] Petrović, D., Marić, F.: Formalizing Analytic Geometries. In: *Automated Deduction in Geometry (ADG 2012)* (2012)
- [10] Boutry, P., Braun, G., Narboux, J.: Formalization of the arithmetization of euclidean plane geometry and applications. *Journal of Symbolic Computation* **90**, 149–168 (2019) <https://doi.org/10.1016/j.jsc.2018.04.007>
- [11] Makarios, T.J.M.: A Mechanical Verification of the Independence of Tarski's Euclidean Axiom (2012) <https://doi.org/10.26686/wgtn.16999462>
- [12] Harrison, J.: A HOL theory of Euclidean space. In: Hurd, J., Melham, T. (eds.) *Theorem Proving in Higher Order Logics, 18th International Conference, TPHOLs 2005. Lecture Notes in Computer Science*, vol. 3603. Springer, Oxford, UK (2005)
- [13] Coghetto, R.: Klein-beltrami model. part i. *Formalized Mathematics* **26**, 21–32 (2018) <https://doi.org/10.2478/forma-2018-0003>
- [14] Marić, F., Petrović, D.: Formalizing complex plane geometry. *Annals of Mathematics and Artificial Intelligence* **74**(3), 271–308 (2015) <https://doi.org/10.1007/s10472-014-9436-4>
- [15] Simić, D., Marić, F., Boutry, P.: Formalization of the poincaré disc model of hyperbolic geometry. *Journal of Automated Reasoning* **65**(1), 31–73 (2021) <https://doi.org/10.1007/s10817-020-09551-2>
- [16] Andréka, H., Németi, I., Madarász, J.X., Székely, G.: On Logical Analysis of Relativity Theories (2011). <https://arxiv.org/abs/1105.0885>
- [17] Kapustin, A.: Is there life beyond Quantum Mechanics? (2013). <https://arxiv.org/abs/1305.0038>

[org/abs/1303.6917](https://doi.org/abs/1303.6917)

- [18] Fleuriot, J.: A combination of geometry theorem proving and nonstandard analysis with application to newton's principia (2001) <https://doi.org/10.1007/978-0-85729-329-9>
- [19] Wenzel, M.: Isabelle, isar - a versatile environment for human readable formal proof documents. (2002). <https://api.semanticscholar.org/CorpusID:1495755>
- [20] Ballarín, C.: Interpretation of locales in isabelle: Theories and proof contexts. In: Mathematical Knowledge Management, MKM, Proceedings, pp. 31–43 (2006)
- [21] Haftmann, F., Wenzel, M.: Constructive type classes in isabelle. In: Altenkirch, T., McBride, C. (eds.) Types for Proofs and Programs, pp. 160–174. Springer, Berlin, Heidelberg (2007)
- [22] Huffman, B., Kunčar, O.: Lifting and Transfer: A Modular Design for Quotients in Isabelle/HOL. In: Gonthier, G., Norrish, M. (eds.) Certified Programs and Proofs, pp. 131–146. Springer, Cham (2013)
- [23] Böhme, S., Nipkow, T.: Sledgehammer: judgement day. In: Proceedings of the 5th International Conference on Automated Reasoning, IJCAR'10, pp. 107–121. Springer, Berlin, Heidelberg (2010). https://doi.org/10.1007/978-3-642-14203-1_9 . https://doi.org/10.1007/978-3-642-14203-1_9